

Design with Hardware Description Languages (HDL)

(DHDL-94952)

Amin Mehranzadeh, Ph.D.

CE Department of
Islamic Azad University of Dezful

mehranzadeh@iaud.ac.ir
Mehran.students@gmail.com

Sequential Code

Amin Mehranzadeh, Ph.D.

CE Department of Islamic Azad University of Dezful

Sequential Code:

- As mentioned before, VHDL code is inherently concurrent. PROCESSES, FUNCTIONS, and PROCEDURES are the only sections of code that are executed sequentially. However, as a whole, any of these blocks is still concurrent with any other statements placed outside it.
- Sequential code is also called behavioral code.

Amin Mehranzadeh, Ph.D.

CE Department of Islamic Azad University of Dezful

Sequential Code:

- The statements discussed in this section are all sequential, that is, allowed only inside PROCESSES, FUNCTIONS, or PROCEDURES. They are: IF, WAIT, CASE, and LOOP.
- VARIABLES are also restricted to be used in sequential code only (that is, inside a PROCESS, FUNCTION, or PROCEDURE). Thus, contrary to a SIGNAL, a VARIABLE can never be global, so its value can not be passed out directly.

Amin Mehranzadeh, Ph.D.

CE Department of Islamic Azad University of Dezful

PROCESS:

- A PROCESS is a sequential section of VHDL code.
- It is characterized by the presence of IF, WAIT, CASE, or LOOP, and by a sensitivity list (except when WAIT is used).
- A PROCESS must be installed in the main code, and is executed every time a signal in the sensitivity list changes (or the condition related to WAIT is fulfilled).

Amin Mehranzadeh, Ph.D.

CE Department of Islamic Azad University of Dezful

PROCESS:

- Its syntax is shown below:

```
[label:] PROCESS (sensitivity list)
  [VARIABLE name type [range] [:= initial_value;]]
BEGIN
  (sequential code)
END PROCESS [label];
```

Notes:


- VARIABLES are optional. If used, they must be declared in the declarative part of the PROCESS (before the word BEGIN, as indicated in the syntax above).
- The initial value is not synthesizable, being only taken into consideration in simulations.
- The use of a label is also optional. Its purpose is to improve code readability. The label can be any word, except VHDL reserved words.

Amin Mehranzadeh, Ph.D.

CE Department of Islamic Azad University of Dezful

Example 1: DFF with Asynchronous Reset #1

- A D-type flip-flop (DFF, figure 6.1) is the most basic building block in sequential logic circuits. In it, the output must copy the input at either the positive or negative transition of the clock signal (rising or falling edge).



Amin Mehranzadeh, Ph.D.

CE Department of Islamic Azad University of Dezful

Example 1: DFF with Asynchronous Reset #1

```

1  -----
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  -----
5  ENTITY dff IS
6      PORT (d, clk, rst: IN STD_LOGIC;
7            q: OUT STD_LOGIC);
8  END dff;
9  -----
10 ARCHITECTURE behavior OF dff IS
11 BEGIN
12     PROCESS (clk, rst)
13     BEGIN
14         IF (rst='1') THEN
15             q <= '0';
16         ELSIF (clk'EVENT AND clk='1') THEN
17             q <= d;
18         END IF;
19     END PROCESS;
20 END behavior;
21 -----

```

Amin Mehranzadeh, Ph.D.

Signals and Variables:

- VHDL has two ways of passing non-static values around: by means of a SIGNAL or by means of a VARIABLE.
- A SIGNAL can be declared in a PACKAGE, ENTITY or ARCHITECTURE (in its declarative part), while a VARIABLE can only be declared inside a piece of sequential code (in a PROCESS, for example).
- Therefore, while the value of the former can be global, the latter is always local.

Amin Mehranzadeh, Ph.D.

CE Department of Islamic Azad University of Dezful

Signals and Variables:

- The value of a VARIABLE can never be passed out of the PROCESS directly; if necessary, then it must be assigned to a SIGNAL. On the other hand, the update of a VARIABLE is immediate, that is, we can promptly count on its new value in the next line of code.
- That is not the case with a SIGNAL (when used in a PROCESS), for its new value is generally only guaranteed to be available after the conclusion of the present run of the PROCESS.

Amin Mehranzadeh, Ph.D.

CE Department of Islamic Azad University of Dezful

IF:

- The syntax of IF is shown below:

```
IF conditions THEN assignments;
ELSIF conditions THEN assignments;
...
ELSE assignments;
END IF;
```

Example:



```
IF (x<y) THEN temp:="11111111";
ELSIF (x=y AND w='0') THEN temp:="11110000";
ELSE temp:=(OTHERS =>'0');
```

Amin Mehranzadeh, Ph.D.

CE Department of Islamic Azad University of Dezful

Example 2: One-digit Counter #1

- The code below implements a progressive 1-digit decimal counter (0 -> 9 -> 0).



Amin Mehranzadeh, Ph.D.

CE Department of Islamic Azad University of Dezful

Example 2: One-digit Counter #1

```


1  -----
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  -----
5  ENTITY counter IS
6      PORT (clk : IN STD_LOGIC;
7            digit : OUT INTEGER RANGE 0 TO 9);
8  END counter;
9  -----
10 ARCHITECTURE counter OF counter IS
11 BEGIN
12     count: PROCESS(clk)
13         VARIABLE temp : INTEGER RANGE 0 TO 10;
14     BEGIN
15         IF (clk'EVENT AND clk='1') THEN
16             temp := temp + 1;
17             IF (temp=10) THEN temp := 0;
18             END IF;
19         END IF;
20         digit <= temp;
21     END PROCESS count;
22 END counter;
23 -----

```

Amin Mehranzadeh, Ph.D.

Example 3: Shift Register

- Figure below shows a 4-bit shift register. The output bit (q) must be four positive clock edges behind the input bit (d). It also contains an asynchronous reset, which must force all flip-flop outputs to '0' when asserted.



Amin Mehranzadeh, Ph.D.

CE Department of Islamic Azad University of Dezful

Example 3: Shift Register

```

1 -----
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  -----
5  ENTITY shiftreg IS
6      GENERIC (n: INTEGER := 4);    -- # of stages
7      PORT (d, clk, rst: IN STD_LOGIC;
8            q: OUT STD_LOGIC);
9  END shiftreg;
10 -----
11 ARCHITECTURE behavior OF shiftreg IS
12     SIGNAL internal: STD_LOGIC_VECTOR (n-1 DOWNTO 0);
13 BEGIN
14     PROCESS (clk, rst)
15     BEGIN
16         IF (rst='1') THEN
17             internal <= (OTHERS => '0');
18         ELSIF (clk'EVENT AND clk='1') THEN
19             internal <= d & internal(internal'LEFT DOWNTO 1);
20         END IF;
21     END PROCESS;
22     q <= internal(0);
23 END behavior;
24 -----

```

Amin Mehranzadeh, Ph.D.

WAIT:

- The operation of WAIT is sometimes similar to that of IF. Moreover, contrary to when IF, CASE, or LOOP are used, the PROCESS cannot have a sensitivity list when WAIT is employed. Its syntax (there are three forms of WAIT) is shown below.

```
WAIT UNTIL signal_condition;
```

```
WAIT ON signal1 [, signal2, ... ];
```

```
WAIT FOR time;
```

Amin Mehranzadeh, Ph.D.

CE Department of Islamic Azad University of Dezful

WAIT UNTIL:

- The WAIT UNTIL statement accepts only one signal, thus being more appropriate for synchronous code than asynchronous. Since the PROCESS has no sensitivity list in this case, WAIT UNTIL must be the first statement in the PROCESS. The PROCESS will be executed every time the condition is met.

Example: 8-bit register with synchronous reset.

```
PROCESS          -- no sensitivity list
BEGIN
    WAIT UNTIL (clk'EVENT AND clk='1');
    IF (rst='1') THEN
        output <= "00000000";
    ELSIF (clk'EVENT AND clk='1') THEN
        output <= input;
    END IF;
END PROCESS;
```

Amin Mehranzadeh, Ph.D.

WAIT ON:

- WAIT ON, on the other hand, accepts multiple signals.
- The PROCESS is put on hold until any of the signals listed changes.
- In the example below, the PROCESS will continue execution whenever a change in rst or clk occurs.

```
PROCESS
BEGIN
    WAIT ON clk, rst;
    IF (rst='1') THEN
        output <= "00000000";
    ELSIF (clk'EVENT AND clk='1') THEN
        output <= input;
    END IF;
END PROCESS;
```

Amin Mehranzadeh, Ph.D.

WAIT FOR:

- Finally, WAIT FOR is intended for simulation only (waveform generation for testbenches).
- Example: WAIT FOR 5ns;

Example 4: DFF with Asynchronous Reset #2

Amin Mehranzadeh, Ph.D.

CE Department of Islamic Azad University of Dezful

Example 4: DFF with Asynchronous Reset #2

- The code below implements the same DFF of example1.
- However, here WAIT ON is used instead of IF only.

```

1  -----
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  -----
5  ENTITY dff IS
6      PORT (d, clk, rst: IN STD_LOGIC;
7            q: OUT STD_LOGIC);
8  END dff;
9  -----
10 ARCHITECTURE dff OF dff IS
11 BEGIN
12     PROCESS
13     BEGIN
14         WAIT ON rst, clk;
15         IF (rst='1') THEN
16             q <= '0';
17         ELSIF (clk'EVENT AND clk='1') THEN
18             q <= d;
19         END IF;
20     END PROCESS;
21 END dff;
22 -----

```

Amin Mehranzadeh, Ph.D.

Example 5: One-digit Counter #2

- The code below implements the same progressive 1-digit decimal counter of example2.
- However, WAIT UNTIL was used instead of IF only.

Amin Mehranzadeh, Ph.D.

```

1  -----
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  -----
5  ENTITY counter IS
6      PORT (clk : IN STD_LOGIC;
7            digit : OUT INTEGER RANGE 0 TO 9);
8  END counter;
9  -----
10 ARCHITECTURE counter OF counter IS
11 BEGIN
12     PROCESS          -- no sensitivity list
13         VARIABLE temp : INTEGER RANGE 0 TO 10;
14     BEGIN
15         WAIT UNTIL (clk'EVENT AND clk='1');
16         temp := temp + 1;
17         IF (temp=10) THEN temp := 0;
18         END IF;
19         digit <= temp;
20     END PROCESS;
21 END counter;
22 -----

```

CASE:

- CASE is another statement intended exclusively for sequential code (along with IF, LOOP, and WAIT). Its syntax is shown below.

```

CASE identifier IS
    WHEN value => assignments;
    WHEN value => assignments;
    ...
END CASE;

```

Example:

```

CASE control IS
    WHEN "00" => x<=a; y<=b;
    WHEN "01" => x<=b; y<=c;
    WHEN OTHERS => x<="0000"; y<="ZZZZ";
END CASE;

```

Amin Mehranzadeh, Ph.D.

CASE:

- The CASE statement (sequential) is very similar to WHEN (combinational).
- Here too all permutations must be tested, so the keyword OTHERS is often helpful.
- Another important keyword is NULL (the counterpart of UNAFFECTED), which should be used when no action is to take place. For example, WHEN OTHERS =>NULL;
- However, CASE allows multiple assignments for each test condition, while WHEN allows only one.

Amin Mehranzadeh, Ph.D.

CE Department of Islamic Azad University of Dezful

CASE:

- Like in the case of WHEN, here too “WHEN value” can take up three forms:

```
WHEN value           -- single value
WHEN value1 to value2 -- range, for enumerated data types
                    -- only
WHEN value1 | value2 | ... -- value1 or value2 or ...
```

Amin Mehranzadeh, Ph.D.

CE Department of Islamic Azad University of Dezful

Example 6: DFF with Asynchronous Reset #3

- The code below implements the same DFF of example1.
- However, here CASE was used instead of IF only.

```


1 -----
2 LIBRARY ieee;                -- Unnecessary declaration,
3                               -- because
4 USE ieee.std_logic_1164.all; -- BIT was used instead of
5                               -- STD_LOGIC
6 -----
7 ENTITY dff IS
8     PORT (d, clk, rst: IN BIT;
9           q: OUT BIT);
10 END dff;
11 -----
12 ARCHITECTURE dff3 OF dff IS
13 BEGIN
14     PROCESS (clk, rst)
15     BEGIN
16         CASE rst IS
17             WHEN '1' => q<='0';
18             WHEN '0' =>
19                 IF (clk'EVENT AND clk='1') THEN
20                     q <= d;
21                 END IF;
22             WHEN OTHERS => NULL; -- Unnecessary, rst is of type
23                                   -- BIT
24         END CASE;
25     END PROCESS;
26 END dff3;
27 -----

```

Amin Mehranzadeh, Ph.D.

Example 7: Two-digit Counter with SSD Output

- The code below implements a progressive 2-digit decimal counter (0 -> 99 -> 0), with external asynchronous reset plus binary-coded decimal (BCD) to seven-segment display (SSD) conversion. Notice that we have chosen the following connection between the
- circuit and the SSD: xabcdefg (that is, the MSB feeds the decimal point, while the LSB feeds segment g).



Amin Mehranzadeh, Ph.D.

Example 7: Two-digit Counter with SSD Output

```

1  -----
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  -----
5  ENTITY counter IS
6      PORT (clk, reset : IN STD_LOGIC;
7            digit1, digit2 : OUT STD_LOGIC_VECTOR (6 DOWNTO 0));
8  END counter;
9  -----
10 ARCHITECTURE counter OF counter IS
11 BEGIN
12     PROCESS(clk, reset)
13         VARIABLE temp1: INTEGER RANGE 0 TO 10;
14         VARIABLE temp2: INTEGER RANGE 0 TO 10;
15     BEGIN
16         ---- counter: -----
17         IF (reset='1') THEN
18             temp1 := 0;
19             temp2 := 0;
20         ELSIF (clk'EVENT AND clk='1') THEN
21             temp1 := temp1 + 1;
22             IF (temp1=10) THEN
23                 temp1 := 0;
24                 temp2 := temp2 + 1;
25                 IF (temp2=10) THEN
26                     temp2 := 0;

```

Amin Mehranzadeh, Ph.D.

Example 7: Two-digit Counter with SSD Output

```

27         END IF;
28     END IF;
29 END IF;
30 ---- BCD to SSD conversion: -----
31 CASE temp1 IS
32     WHEN 0 => digit1 <= "1111110"; --7E
33     WHEN 1 => digit1 <= "0110000"; --30
34     WHEN 2 => digit1 <= "1101101"; --6D
35     WHEN 3 => digit1 <= "1111001"; --79
36     WHEN 4 => digit1 <= "0110011"; --33
37     WHEN 5 => digit1 <= "1011011"; --5B
38     WHEN 6 => digit1 <= "1011111"; --5F
39     WHEN 7 => digit1 <= "1110000"; --70
40     WHEN 8 => digit1 <= "1111111"; --7F
41     WHEN 9 => digit1 <= "1111011"; --7B
42     WHEN OTHERS => NULL;
43 END CASE;
44 CASE temp2 IS
45     WHEN 0 => digit2 <= "1111110"; --7E
46     WHEN 1 => digit2 <= "0110000"; --30
47     WHEN 2 => digit2 <= "1101101"; --6D
48     WHEN 3 => digit2 <= "1111001"; --79
49     WHEN 4 => digit2 <= "0110011"; --33
50     WHEN 5 => digit2 <= "1011011"; --5B
51     WHEN 6 => digit2 <= "1011111"; --5F
52     WHEN 7 => digit2 <= "1110000"; --70
53     WHEN 8 => digit2 <= "1111111"; --7F
54     WHEN 9 => digit2 <= "1111011"; --7B
55     WHEN OTHERS => NULL;
56 END CASE;
57 END PROCESS;
58 END counter;
59 -----

```

Amin Mehranzadeh, Ph.D.

LOOP:

- As the name says, LOOP is useful when a piece of code must be instantiated several times. There are several ways of using LOOP, as shown in the syntaxes below.

FOR / LOOP: The loop is repeated a fixed number of times.

```
[label:] FOR identifier IN range LOOP
  (sequential statements)
END LOOP [label];
```

Note:

One important remark regarding FOR / LOOP (similar to that made for GENERATE) is that both limits of the range must be static.

Example of FOR / LOOP:

```
FOR i IN 0 TO 5 LOOP
  x(i) <= enable AND w(i+2);
  y(0, i) <= w(i);
END LOOP;
```

Amin Mehranzadeh, Ph.D.

CE Department of Islamic Azad University of Dezful

LOOP:

WHILE / LOOP: The loop is repeated until a condition no longer holds.

```
[label:] WHILE condition LOOP
  (sequential statements)
END LOOP [label];
```

Example of WHILE / LOOP: In this example, LOOP will keep repeating while $i < 10$.

```
WHILE (i < 10) LOOP
  WAIT UNTIL clk'EVENT AND clk='1';
  (other statements)
END LOOP;
```

Amin Mehranzadeh, Ph.D.

CE Department of Islamic Azad University of Dezful

LOOP:

EXIT: Used for ending the loop.

```
[label:] EXIT [label] [WHEN condition];
```

Example with EXIT: In this case, the loop will end as soon as a value different from '0' is found in the data vector.

```
FOR i IN data'RANGE LOOP
  CASE data(i) IS
    WHEN '0' => count:=count+1;
    WHEN OTHERS => EXIT;
  END CASE;
END LOOP;
```

Amin Mehranzadeh, Ph.D.

CE Department of Islamic Azad University of Dezful

LOOP:

NEXT: Used for skipping loop steps.

```
[label:] NEXT [loop_label] [WHEN condition];
```

Example with NEXT: In the example below, NEXT causes LOOP to skip one iteration when i = skip.


```
FOR i IN 0 TO 15 LOOP
  NEXT WHEN i=skip;    -- jumps to next iteration
  (...)
END LOOP;
```

Amin Mehranzadeh, Ph.D.

CE Department of Islamic Azad University of Dezful

Example 8: Carry Ripple Adder

Figure below shows an 8-bit unsigned carry ripple adder.



Each section of the latter diagram is a full-adder unit. Thus its outputs can be computed by means of:

$$s_j = a_j \text{ XOR } b_j \text{ XOR } c_j$$

$$c_{j+1} = (a_j \text{ AND } b_j) \text{ OR } (a_j \text{ AND } c_j) \text{ OR } (b_j \text{ AND } c_j)$$

Amin Mehranzadeh, Ph.D.

CE Department of Islamic Azad University of Dezful

Example 8: Carry Ripple Adder

```

1  ----- Solution 1: Generic, with VECTORS -----
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  -----
5  ENTITY adder IS
6      GENERIC (length : INTEGER := 8);
7      PORT ( a, b: IN STD_LOGIC_VECTOR (length-1 DOWNTO 0);
8            cin: IN STD_LOGIC;
9            s: OUT STD_LOGIC_VECTOR (length-1 DOWNTO 0);
10           cout: OUT STD_LOGIC);
11 END adder;
12 -----
13 ARCHITECTURE adder OF adder IS
14 BEGIN
15     PROCESS (a, b, cin)
16         VARIABLE carry : STD_LOGIC_VECTOR (length DOWNTO 0);
17     BEGIN
18         carry(0) := cin;
19         FOR i IN 0 TO length-1 LOOP
20             s(i) <= a(i) XOR b(i) XOR carry(i);
21             carry(i+1) := (a(i) AND b(i)) OR (a(i) AND
22                             carry(i)) OR (b(i) AND carry(i));
23         END LOOP;
24         cout <= carry(length);
25     END PROCESS;
26 END adder;
27 -----

```

Amin Mehranzadeh, Ph.D.

Example 8: Carry Ripple Adder

```

1  ---- Solution 2: non-generic, with INTEGERS ----
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  -----
5  ENTITY adder IS
6      PORT ( a, b: IN INTEGER RANGE 0 TO 255;
7            c0: IN STD_LOGIC;
8            s: OUT INTEGER RANGE 0 TO 255;
9            c8: OUT STD_LOGIC);
10 END adder;
11 -----
12 ARCHITECTURE adder OF adder IS
13 BEGIN
14     PROCESS (a, b, c0)
15         VARIABLE temp : INTEGER RANGE 0 TO 511;
16     BEGIN
17         IF (c0='1') THEN temp:=1;
18         ELSE temp:=0;
19         END IF;
20         temp := a + b + temp;
21         IF (temp > 255) THEN
22             c8 <= '1';
23             temp := temp---256;
24         ELSE c8 <= '0';
25         END IF;
26         s <= temp;
27     END PROCESS;
28 END adder;
29 -----


```

Amin Mehranzadeh, Ph.D.

Example 9: (Simple Barrel Shifter)

the circuit must shift the input vector (of size 8) either 0 or 1 position to the left. When actually shifted (shift = 1), the LSB bit must be filled with '0' (shown in the bottom left corner of the diagram). If shift = 0, then $outp = inp$; if shift = 1, then $outp(0) = '0'$ and $outp(i) = inp(i - 1)$, for $1 \leq i \leq 7$.

Amin Mehranzadeh, Ph.D.



Example 9: (Simple Barrel Shifter)

Amin Mehranzadeh, Ph.D.

```

1  -----
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  -----
5  ENTITY barrel IS
6      GENERIC (n: INTEGER := 8);
7      PORT ( inp: IN STD_LOGIC_VECTOR (n-1 DOWNT0 0);
8            shift: IN INTEGER RANGE 0 TO 1;
9            outp: OUT STD_LOGIC_VECTOR (n-1 DOWNT0 0));
10 END barrel;
11 -----
12 ARCHITECTURE RTL OF barrel IS
13 BEGIN
14     PROCESS (inp, shift)
15     BEGIN
16         IF (shift=0) THEN
17             outp <= inp;
18         ELSE
19             outp(0) <= '0';
20             FOR i IN 1 TO inp'HIGH LOOP
21                 outp(i) <= inp(i-1);
22             END LOOP;
23         END IF;
24     END PROCESS;
25 END RTL;
26 -----

```

Example 10: (Leading Zeros)

The design below counts the number of leading zeros in a binary vector, starting from the left end. In this example, the loop will end as soon as a '1' is found in the data vector. Therefore, it is appropriate for counting the number of zeros that precedes the first one.

Amin Mehranzadeh, Ph.D.

```

1  -----
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  -----
5  ENTITY LeadingZeros IS
6      PORT ( data: IN STD_LOGIC_VECTOR (7 DOWNT0 0);
7            zeros: OUT INTEGER RANGE 0 TO 8);
8  END LeadingZeros;
9  -----
10 ARCHITECTURE behavior OF LeadingZeros IS
11 BEGIN
12     PROCESS (data)
13         VARIABLE count: INTEGER RANGE 0 TO 8;
14     BEGIN
15         count := 0;
16         FOR i IN data'RANGE LOOP
17             CASE data(i) IS
18                 WHEN '0' => count := count + 1;
19                 WHEN OTHERS => EXIT;
20             END CASE;
21         END LOOP;
22         zeros <= count;
23     END PROCESS;
24 END behavior;
25 -----

```